

GUIA TÉCNICO DO PROJETO

Sistema de Acompanhamento de Vendas, Metas e Comissões

Excel Online • janeiro a abril de 2026

Finalidade

Registrar a arquitetura do arquivo, todas as fórmulas identificadas, o objetivo de cada cálculo e o procedimento para acrescentar novos meses sem Power Query.

Documento elaborado a partir do arquivo ODS final enviado em setembro de 2026.

1. Visão geral

O projeto consolida quatro bases mensais de vendas, enriquece os registros com custos e equipes, calcula faturamento, margem, metas e comissões, valida a qualidade dos dados e apresenta os resultados em análises e dashboard. Como a solução foi construída no Excel Online sem Power Query, a entrada de um novo mês exige a ampliação manual de alguns intervalos.

Regra central de faturamento

Somente vendas com Status_Venda igual a “Concluída” entram no faturamento, no custo, na margem e no cálculo de comissão. Cancelamentos e devoluções permanecem na base para análise, mas contribuem com zero.

2. Estrutura das abas

Aba	Objetivo
Vendas_Jan a Vendas_Abr	Bases mensais originais com 42 registros por mês.
Base_Consolidada	Empilhamento das bases e cálculos por venda.
Cadastro_Vendedores	Nome, região, equipe e admissão de cada vendedor.
Cadastro_Produtos	Produto, categoria, custo unitário e preço de catálogo.
Cadastro_Metas	Meta mensal por vendedor.
Faixas_Comissao	Percentual de comissão por faixa de atingimento.
Comissoes	Resumo mensal por vendedor e aplicação das regras comerciais.
Ranking	Consolidação por vendedor e ordenação por faturamento.
Validacao	Testes de integridade, consistência e reconciliação financeira.
Analise_Vendedores	Tabela dinâmica e gráfico de faturamento versus meta.
Analise_Mensal	Evolução mensal de faturamento e meta.
Analise_Produtos	Faturamento por produto e filtros de mês/categoria.
Analise_Canais	Faturamento por região e canal.
Dashboard	Indicadores e gráficos consolidados.

3. Mapa das fórmulas

Aba	Células / colunas	Funções principais
Base_Consolidada	A2 e O:X	EMPILHARV, SE, SEERRO, PROCX, DATA, ANO, MÊS
Comissoes	C:J	PROCX, SOMASES, SEERRO, SES
Ranking	A e C:H	ORDEM.EQ, PROCX, SOMASE, SEERRO
Validacao	B2:B14	CONT.VALORES, ÚNICO, CONT.SE, SOMA, SOMARPRODUTO
Dashboard	B5:F5	SOMA, SEERRO

4. Fórmulas da Base_Consolidada

A fórmula de A2 gera uma matriz dinâmica nas colunas A:N. As fórmulas de O:X são preenchidas para cada registro, atualmente até a linha 169.

A2 — Consolidação das bases mensais — Reúne as 42 linhas de cada mês em uma única base de 168 vendas.

```
=EMPILHARV(Vendas_Jan!A2:N43;Vendas_Fev!A2:N43;Vendas_Mar!A2:N43;Vendas_Abr!A2:N43)
```

Ao incluir um novo mês, acrescente a nova referência antes do parêntese final. A fórmula deve existir somente em A2; o resultado se espalha automaticamente por A:N.

O2 — Valor_Bruto — Calcula o valor antes do desconto.

```
=J2*K2
```

Quantidade × Preço unitário.

P2 — Valor_Desconto — Transforma o percentual de desconto em valor monetário.

```
=O2*L2
```

Valor bruto × Desconto percentual.

Q2 — Valor_Liquido — Calcula o valor líquido da venda após o desconto.

```
=O2-P2
```

R2 — Valor_Considerado — Define o faturamento válido.

```
=SE(M2="Concluída";Q2;0)
```

Vendas canceladas ou devolvidas recebem zero e não entram nos indicadores comerciais.

S2 — Custo_Unitario — Busca o custo unitário pelo ID do produto.

```
=SEERRO(PROCX(E2;Cadastro_Produtos!$A$2:$A$11;Cadastro_Produtos!$D$2:$D$11);0)
```

SEERRO devolve zero quando o produto não é localizado; a aba Validacao detecta esses casos.

T2 — Custo_Total — Calcula o custo somente das vendas concluídas.

```
=SE(M2="Concluída";J2*S2;0)
```

Quantidade × Custo unitário; cancelamentos e devoluções recebem zero.

U2 — Margem — Calcula a margem bruta monetária da venda.

```
=R2-T2
```

Faturamento considerado – Custo total.

V2 — Margem_Percentual — Mede a margem como percentual do faturamento.

```
=SEERRO(U2/R2;0)
```

O tratamento de erro evita divisão por zero em vendas não consideradas.

W2 — Mes — Cria uma data padronizada para representar o mês da venda.

```
=DATA(ANO(B2);MÊS(B2);1)
```

O dia é sempre 1; isso permite cruzar vendas, metas e comissões por mês.

X2 — Equipe — Busca a equipe do vendedor.

```
=SEERRO(PROCX(C2;Cadastro_Vendedores!$A$2:$A$9;Cadastro_Vendedores!$D$2:$D$9);"Não encontrado")
```

A mensagem “Não encontrado” alimenta a verificação de qualidade.

5. Fórmulas da aba Comissoes

Cada linha representa a combinação de um mês e um vendedor. Existem 32 combinações: oito vendedores × quatro meses. As fórmulas abaixo começam na linha 2 e são preenchidas até a linha 33.

C2 — Nome_Vendedor — Converte o ID do vendedor em nome.

```
=SEERRO(PROCX(B2;Cadastro_Vendedores!$A$2:$A$9;Cadastro_Vendedores!$B$2:$B$9);"Não encontrado")
```

D2 — Equipe — Identifica a equipe do vendedor.

```
=SEERRO(PROCX(B2;Cadastro_Vendedores!$A$2:$A$9;Cadastro_Vendedores!$D$2:$D$9);"Não encontrado")
```

E2 — Faturamento — Soma o faturamento válido do vendedor no mês.

```
=SOMASES(Base_Consolidada!$R$2:$R$169;Base_Consolidada!$C$2:$C$169;B2;Base_Consolidada!$W$2:$W$169;A2)
```

Crítérios: ID do vendedor e mês. A coluna somada é Valor_Considerado, portanto cancelamentos e devoluções já estão excluídos.

F2 — Meta — Busca a meta daquele vendedor naquele mês por agregação condicional.

```
=SOMASES(Cadastro_Metas!$C$2:$C$33;Cadastro_Metas!$A$2:$A$33;A2;Cadastro_Metas!$B$2:$B$33;B2)
```

G2 — Atingimento — Calcula quanto da meta foi atingido.

```
=SEERRO(E2/F2;0)
```

Exemplo: 0,80 corresponde a 80%. Retorna zero se não houver meta válida.

H2 — Percentual_Comissao — Seleciona a taxa de comissão da maior faixa menor ou igual ao atingimento.

```
=SEERRO(PROCX(G2;Faixas_Comissao!$A$2:$A$5;Faixas_Comissao!$B$2:$B$5;;-1);0)
```

Faixas atuais: 0%→1%; 80%→2%; 100%→3%; 120%→5%. O modo de correspondência -1 faz a busca aproximada para baixo.

I2 — Valor_Comissao — Calcula o valor monetário da comissão.

```
=E2*H2
```

Faturamento do vendedor × Percentual de comissão aplicável.

J2 — Status_Meta — Classifica o desempenho comercial em quatro níveis.

```
=SES(G2>=1,2;"Superou a meta";G2>=1;"Meta atingida";G2>=0,8;"Próximo da meta";VERDADEIRO;"Abaixo da meta")
```

A ordem das condições é importante: a fórmula testa primeiro a maior faixa.

6. Fórmulas da aba Ranking

O ranking consolida os quatro meses por vendedor. As fórmulas são preenchidas da linha 2 até a linha 9.

A2 — Posicao — Posiciona o vendedor do maior para o menor faturamento.

```
=ORDEM.EQ(D2;$D$2:$D$9;0)
```

O argumento 0 define ordem decrescente.

C2 — Nome_Vendedor — Busca o nome pelo ID.

```
=SEERRO(PROCX(B2;Cadastro_Vendedores!$A$2:$A$9;Cadastro_Vendedores!$B$2:$B$9);"Não encontrado")
```

D2 — Faturamento — Soma o faturamento do vendedor nos quatro meses.

```
=SOMASE(Comissoes!$B$2:$B$33;B2;Comissoes!$E$2:$E$33)
```

E2 — Meta — Soma as metas mensais do vendedor.

```
=SOMASE(Comissoes!$B$2:$B$33;B2;Comissoes!$F$2:$F$33)
```

F2 — Atingimento — Calcula o atingimento acumulado da meta.

```
=SEERRO(D2/E2;0)
```

G2 — Margem — Soma a margem das vendas do vendedor.

```
=SOMASE(Base_Consolidada!$C$2:$C$169;B2;Base_Consolidada!$U$2:$U$169)
```

H2 — Comissao — Soma as comissões mensais do vendedor.

```
=SOMASE(Comissoes!$B$2:$B$33;B2;Comissoes!$I$2:$I$33)
```

7. Fórmulas da aba Validacao

Estas fórmulas funcionam como testes de qualidade e reconciliação. Abaixo estão as versões recomendadas para a base atual, com 168 registros até a linha 169.

B2 — Total de registros — Conta os IDs preenchidos na base.

```
=CONT.VALORES(Base_Consolidada!A2:A169)
```

B3 — Vendas únicas — Conta quantos IDs distintos existem.

```
=CONT.VALORES(ÚNICO(Base_Consolidada!A2:A169))
```

B4 — IDs duplicados — Calcula a diferença entre registros totais e IDs únicos.

```
=B2-B3
```

Resultado esperado: 0.

B5 — Produtos não encontrados — Conta buscas de custo que retornaram zero.

```
=CONT.SE(Base_Consolidada!S2:S169;0)
```

Resultado esperado: 0, considerando que nenhum produto válido possui custo unitário zero.

B6 — Vendedores não encontrados — Conta vendedores sem correspondência no cadastro.

```
=CONT.SE(Base_Consolidada!X2:X169;"Não encontrado")
```

Resultado esperado: 0.

B7 — Faturamento na base — Soma o faturamento válido diretamente na base consolidada.

```
=SOMA(Base_Consolidada!R2:R169)
```

B8 — Faturamento nas comissões — Soma o faturamento consolidado na aba de comissões.

```
=SOMA(Comissoes!E2:E33)
```

B9 — Diferença — Reconcilia os dois cálculos de faturamento.

```
=B7-B8
```

Resultado esperado: 0.

B10 — Quantidades inválidas — Conta vendas com quantidade igual ou inferior a zero.

```
=CONT.SE(Base_Consolidada!J2:J169;"<=0")
```

Resultado esperado: 0.

B11 — Preços abaixo do custo — Conta produtos vendidos por preço unitário inferior ao custo.

```
=SOMARPRODUTO(--(Base_Consolidada!K2:K169<Base_Consolidada!S2:S169))
```

Resultado esperado no arquivo atual: 0.

B12 — Descontos inválidos — Conta percentuais de desconto menores que 0% ou maiores que 100%.

```
=CONT.SE(Base_Consolidada!L2:L169;"<0")+CONT.SE(Base_Consolidada!L2:L169;">1")
```

Resultado esperado: 0.

B13 — Status inválidos — Conta status diferentes dos três valores aceitos.

```
=CONT.VALORES(Base_Consolidada!M2:M169)-CONT.SE(Base_Consolidada!M2:M169;"Concluída")-CONT.SE(Base_Consolidada!M2:M169;"Cancelada")-CONT.SE(Base_Consolidada!M2:M169;"Devolvida")
```

Resultado esperado: 0.

B14 — Datas fora do período — Conta datas anteriores a janeiro ou posteriores a abril de 2026.

```
=CONT.SE(Base_Consolidada!B2:B169;"<"&DATA(2026;1;1))+CONT.SE(Base_Consolidada!B2:B169;">"&DATA(2026;4;30))
```

Ao adicionar outro mês, altere a data final desta fórmula.

8. Fórmulas do Dashboard

B5 — Faturamento — Apresenta o faturamento total dos quatro meses.

```
=SOMA(Comissoes!E2:E33)
```

C5 — Meta — Apresenta a meta total dos quatro meses.

```
=SOMA(Comissoes!F2:F33)
```

D5 — Atingimento — Calcula o percentual consolidado da meta atingida.

```
=SEERRO(B5/C5;0)
```

A fórmula recomendada divide Faturamento por Meta.

E5 — Comissões — Apresenta o total de comissões calculadas.

```
=SOMA(Comissoes!I2:I33)
```

F5 — Margem — Apresenta a margem total das vendas concluídas.

```
=SOMA(Base_Consolidada!U2:U169)
```


9. Tabelas dinâmicas, gráficos e filtros

As abas Analise_Vendedores, Analise_Mensal, Analise_Produtos e Analise_Canais não possuem fórmulas de célula no arquivo exportado. Seus resultados são produzidos por tabelas dinâmicas e alimentam os gráficos do Dashboard.

Análise	Dimensões e métricas	Filtros
Vendedores	Linhas: vendedor; Valores: faturamento e meta	Mês e equipe
Mensal	Linhas: mês; Valores: faturamento e meta	Filtro da própria tabela
Produtos	Linhas: produto; Valor: Valor_Considerado	Mês e categoria
Canais	Linhas: região; Colunas: canal; Valor: Valor_Considerado	Mês

Segmentações

As segmentações permanecem nas abas de análise devido às limitações de movimentação e conexão do Excel Online. Elas demonstram interatividade mesmo sem estarem posicionadas no Dashboard.

10. Fluxo para adicionar um novo mês

Etapas	Ação
1	Criar a nova aba mensal com as mesmas 14 colunas e 42 registros, sem alterar nomes ou ordem dos cabeçalhos.
2	Acrescentar a nova aba à fórmula EMPILHARV em Base_Consolidada!A2.
3	Estender as fórmulas de O:X até a última linha da base consolidada.
4	Adicionar uma meta para cada vendedor em Cadastro_Metas.
5	Adicionar uma linha por vendedor na aba Comissoes e preencher as fórmulas.
6	Ampliar os intervalos das fórmulas em Comissoes, Ranking, Validacao e Dashboard.
7	Atualizar a data final do teste "Datas fora do período".
8	Executar Dados → Atualizar Tudo e conferir todas as tabelas dinâmicas e gráficos.
9	Validar: registros = vendas únicas; duplicados e demais erros = 0; diferença de faturamento = 0.
10	Atualizar o subtítulo e os textos de insights do Dashboard.

11. Glossário das funções

Função	Objetivo no projeto
EMPILHARV	Une intervalos verticalmente em uma matriz dinâmica.
PROCX	Procura um valor em uma coluna e retorna o correspondente de outra coluna.

Função	Objetivo no projeto
SEERRO	Substitui erros por um resultado controlado.
SE / SES	Aplica regras condicionais simples ou múltiplas.
SOMASE / SOMASES	Soma valores conforme um ou vários critérios.
CONT.SE	Conta células que atendem a uma condição.
CONT.VALORES	Conta células não vazias.
ÚNICO	Retorna apenas valores distintos de um intervalo.
SOMARPRODUTO	Avalia e soma operações entre matrizes; aqui, conta comparações verdadeiras.
ORDEM.EQ	Retorna a posição de um valor em um ranking.
DATA / ANO / MÊS	Constroem uma data mensal padronizada a partir da data da venda.

12. Critérios de conclusão

- 168 registros e 168 IDs únicos na base atual.
- Zero IDs duplicados, produtos ou vendedores não encontrados e valores inválidos.
- Faturamento da base igual ao faturamento da aba Comissoes.
- Abril visível e com valores diferentes de zero nas quatro análises e no Dashboard.
- Indicadores do Dashboard atualizados e insights coerentes com janeiro a abril.

13. Limitações do Excel Online

Este projeto foi desenvolvido utilizando a versão gratuita do Excel Online, que possui algumas limitações em comparação com a versão desktop.

Como o Power Query não estava disponível, a solução exigiu:

- Inclusão manual de cada nova aba mensal;
- Uso da função EMPILHARV para consolidar as bases;
- Extensão manual das colunas calculadas;
- Atualização dos intervalos fixos das fórmulas;
- Conferência manual das fontes das tabelas dinâmicas;
- Manutenção das segmentações nas abas de análise devido às limitações da interface.

Como o projeto seria aprimorado no Excel Desktop

Com acesso à versão desktop do Excel, o Power Query poderia ser utilizado para:

- Importar automaticamente os arquivos mensais de uma pasta;
- Combinar novos arquivos de vendas durante a atualização;
- Padronizar nomes de colunas e tipos de dados;
- Remover duplicidades e linhas vazias;

- Identificar valores inválidos, ausentes ou inconsistentes;
- Mesclar as vendas com os cadastros de produtos e vendedores;
- Criar etapas de transformação documentadas e reutilizáveis;
- Gerar uma base analítica tratada para alimentar as análises.

Assim, ao adicionar um novo arquivo mensal à pasta, seria necessário apenas utilizar Dados → Atualizar Tudo. Isso eliminaria grande parte dos ajustes manuais, reduziria o risco de erros e tornaria a solução mais escalável.